
program Driver

  !-----------------------------------------

  !======= Declarations =========

  implicit none

  integer :: n, i, ierr

  double precision, allocatable :: u(:), v(:)

  double precision :: sum

  namelist /inputs/ n

  !======= Internals ============ 

  ! read vector length from file 'input.txt'

  open(100,file='input.txt',form='formatted')

  read(100,inputs)

  close(100)

  ! output run information to screen

  write(*,'(A,i10)') '  vector length = ',n

      

  ! allocate arrays u and v

  allocate(u(1:n))

  allocate(v(1:n))

  ! fill in vectors u and v
  do i=1,n,1
     u(i) = i
     v(i) = n-i+1
  enddo

  ! compute dot-product
  sum = 0.d0
  do i=1,n,1
     sum = sum + u(i)*v(i)
  enddo

  ! output result to screen
  write(*,'(A,es22.15)') ‘dot product =',sum
  
  ! output info & result to file 'output.txt'
  open(10,file='output.txt',form='formatted')
  write(10,'(A,i10)') ' n =',n
  write(10,'(A,es22.15)') ' u dot v =',sum
  close(10)
  
  ! free arrays u and v
  deallocate(u,v)
      
end program Driver
!-----------------------------------------



Fortran recognizes the alphanumeric characters:

◦ A B C D E F G H I J K L M N O P Q R S T U V W X Y Z _

◦ a b c d e f g h i j k l m n o p q r s t u v w x y

◦ 0 1 2 3 4 5 6 7 8 9

◦ = + - * / ( ) , . ’ : ! ” % & ; < > ? $     

◦ [? and $ are unused]

Names of variables, subroutines, functions, etc., must begin with a letter.  
These may include letters, numbers, and _, cannot include spaces, and are 
case-independent (i.e. Name == nAmE).

◦ Good names: max_norm, TotalEnergy, L1Norm

◦ Bad names: 2norm, dot product, _matrix, x-velocity



Arithmetic Operators:

◦ +

◦ -

◦ *

◦ /

◦ **  [^ in Matlab]

Relational Operators:

◦ <, >

◦ <=, >=

◦ ==

◦ /=  [~= in Matlab]

 Logical Operators:

◦ .and.  [&& in Matlab]

◦ .or.  [|| in Matlab]

◦ .not.  [~ in Matlab]

 Special characters:  

◦ comments  !  [% in Matlab]

◦ structure element %           
[. in Matlab]

◦ line continuation &  […  in Matlab]



In the rest of this lecture, we will strive to use the following notation:

Italicized  words will denote items that you may rename

[bracketed]  words will denote optional items

...   will denote areas for commands (or Matlab line continuation)

monospaced font will be used to denote code segments

green words in code blocks will denote comments

Every effort will be made to compare/contrast Fortran commands, 
syntax, and usage with Matlab.  If at any point you are unclear about 
something, please stop me to ask a question.

In working with codes, I suggest turning on syntax highlighting in your 
text editor (in emacs use the ‘Options’ menu), which will colorize 
keywords to make coding easier.



A key improvement in F90 over F77 is the move to allow increased 
flexibility in writing source code programs.

This transition was from so-called “fixed-source form” to “free-source 
form”.

To retain backwards-compatibility, F90 compilers still allow fixed-
source form, but compiler manufacturers (and the Fortran standards 
board) recommend that new codes use free-form.

Fixed-source form is a relic of Fortran’s heritage, when it was 
programmed into punch cards to be fed into a computer.

[Show fixed vs free form example – dot_prod.f vs dot_prod.f90]



Matlab
Script file:

       ...

Fortran
 Main routine:

program program_name 

  ...

end [program [program_name]]

In general, Fortran programs are structured similarly to Matlab programs, 
except that since Fortran is a compiled (Matlab is interpreted), there is no 
command prompt, so all code must be included in a main program, 
subroutines and functions (not typed at the prompt).  

As a result, the code may be written using any text editor of your choosing (I 
like emacs), and you must tell the Fortran compiler that the file contains 
the main program with the “program” keyword.



Matlab

Function without return variables:

function fname(arg1, arg2)

    ...

return

Fortran

 Subroutine without return variables:

subroutine sname(arg1, arg2)

    ...

  return

end [subroutine [sname]]

Routines that take input arguments but do not return anything (e.g. that print 
things to the screen or to an output file), are almost identical between 
Matlab and Fortran, except that ‘function’ is replaced with the keyword 
‘subroutine’, and the file must end with the keyword ‘end’.



Matlab
Single-output function:

function y = fname(arg1, arg2)

  ...

  y = ...

  return

Fortran
 Single-output function:

function fname(arg1, arg2)

  integer :: fname

  ...

  fname = ...

  return

end [function [fname]]

Functions with only a single output are almost identical, except that the 
Fortran function output is given by the function name.  Also, as we’ll 
discuss soon, the data type of the return argument must be specified in 
Fortran.



Matlab:  “call by value”
Multiple-output function:

function [x,y] = fname(a)

  ...

  x = ...

  y = ...

  

  return

Fortran:  “call by reference”
 Multiple-output subroutine:

subroutine sname(a, x, y)

  ...

    x = ...

  y = ...

return

end [subroutine [sname]]

Functions with multiple outputs must be handled differently, since Fortran 
does not allow you to specify different names for function outputs.

The solution highlights a significant difference between Matlab and Fortran.  
While Matlab makes copies of input arguments to use locally in the 
function, Fortran will operate on the input arguments directly (more on this 
later).



A data type is the computer’s format for representing a number, character, or 
memory reference using a given number of bits.  

Matlab:

◦ All variables default to a double-precision matrix, unless otherwise specified.

◦ Variables may be declared anywhere in a program.  

Fortran:

◦ We must specify the data type (and size) for every variable.

◦ Variables must be declared at the beginning of a program or subroutine 
(header block), before any computations are performed. 

◦ Since arguments are passed by reference, they may be changed by the 
subroutine prior to returning them to the calling program.

◦   

Arguments may be declared as input, output, or both by a Fortran subroutine, 
to aid in ownership of the data.  This can also speed up program execution 
since the compiler can take advantage of the additional information.



Fortran has two default kinds of integers:

32-bit integers use 32 bits (4 bytes) of memory:
◦ Declared as:  integer, integer(kind=4), integer(4) or integer*4
◦ IEEE min: -2,147,483,647
◦ IEEE max:  2,147,483,648

64-bit integers use 64 bits (8 bytes) of memory:
◦ Declared as: integer(kind=8), integer(8) or integer*8
◦ IEEE min: -9,223,372,036,854,775,807
◦ IEEE max:  9,223,372,036,854,775,807

Be careful with integer arithmetic:  In order to achieve an integer result, 
non-integer outputs must be rounded (the default is usually to floor toward 0):

1 / 2 = 0



Real-valued numbers have three main types in Fortran:

 32-bit:  declared as  real, real(kind=4), real(4) or real*4

◦ ~7 digits of accuracy

◦ IEEE max:  ~3.4*1038

64-bit:  declared as  double precision, real(kind=8), real(8) or real*8

◦ ~16 digits of accuracy

◦ IEEE max:  ~1.8*10308

128-bit:  declared as  real(kind=16), real(16) or real*16

◦ ~34 digits of accuracy

◦ IEEE max:  ~1.2*104932

Complex numbers combine 2 real numbers together (don’t exist in C/C+
+): 

 64-bit (pair of 32-bit real numbers)

◦ Declared as  complex(kind=4), complex(4) or complex*4

 128-bit (pair of 64-bit reals)

◦ Declared as  complex(kind=8), complex(8) or complex*8



Characters:

character(80), character(len=80): 

◦ defines a variable that holds up to 80 characters

Allow comparison operations: <, <=, >=, >, ==, /=, that perform alphabetical 
comparisons  (e.g.  c > b)

Allows the concatenation operator  //  that combines two character variables,     
  (e.g.   ‘sand’ // ‘wich’  results in ‘sandwich’)

Logical variables:

logical, logical(kind=4), logical(4)
◦ 32-bit logical (Fortran default, to match with C/C++)
◦ value is either .true. or .false.

logical(kind=bit), logical(1)
◦ 1-bit logical
◦ more compact form in memory (not default)



Be careful with constants, since the compiler does not know the intended type.

1/3  ≠  1.0/3.0  ≠  1.d0/3.d0

You can specify the kind of variable to be specific:
Integer: 1, 2, 3 are compiler-dependent:
◦ 1_4 is a 32-bit representation of 1
◦ 2_8 is a 64-bit representation of 2
Real numbers can use an underscore or an ‘e’ or ‘d’:  
◦ 32-bit versions of 1 include:  1.0,  1.,  1.e0, and  1.0_4
◦ 64-bit versions of 1 include:  1.d0  and  1.0_8
Complex:  
◦ 32-bit versions of  2+i  include: (2.0, 1.)  and  (0.2e1, 1._4)
◦ 64-bit versions of  2+i  include: (2._8, 1.d0)  and  (20.0d-1, 1._8)
Logical:  
◦ .true. represents “true” with a 32-bit logical
◦ .true._bit represents “true” with a 1-bit logical



We can declare an array of entries of any intrinsic (or user-defined) type:

1. real(4) a(10)

2. integer :: b(-1:8), c(2:11)

3. complex(8), dimension(10) :: d, e, f

All of these declare arrays of length 10 of the appropriate type.  

The second example shows that we can start arrays with whatever index we 
want.  If left unspecified (as in the first one), the default starting index is 1.

We typically to use the :: notation to differentiate the type and attributes from 
the variables.

These declarations must be the first thing inside of a Fortran program, function 
or subroutine, before any computations.  We call this grouping of declarations 
the “header block”.



Like Matlab, Fortran allows use of multi-dimensional arrays (maximum rank 7):

double precision :: M(2,4), N(4,3,2), P(-3:10,2,100,5:9,8)         
   declares:

◦ A 2D array M with 2 rows & 4 columns, 
◦ a 3D array N of size 4x3x2, and 
◦ a 5D array P of size 14x2x100x5x8.

Fortran uses column-major ordering: in memory, the first indices run fastest.

M = (/ 1,5,2,6,3,7,4,8 /)  gives the same as Matlab’s                                   
M = [1,2,3,4;5,6,7,8]          where the shape of M has been defined already

We access array elements like in Matlab:  M(1,4) gives the entry of M in row 1 
and column 4.

We access sub-array elements like in Matlab as well:  M(2:3,:) gives a sub-
matrix of M containing rows 2 and 3 (and all columns).



As of Fortran90, arrays may be allocated/deallocated dynamically (prior to F90, 
the size of all arrays had to be hard-coded in the program).

This previous deficiency is the main reason Fortran has such a bad reputation in 
the CS community (many do not know it has been remedied for over 20 years).

To declare an array that uses dynamic memory allocation, it must be given the 
‘allocatable’ attribute when it is declared:

real, allocatable, dimension(:,:,:) :: Q     ! declare Q
allocate(Q(1:2,1:5,0:99))                    ! allocate Q
...         ! use Q
deallocate(Q)                                ! free Q



Like Matlab, Fortran allows whole-array operations:

integer, allocatable :: P(:), Q(:)      ! declaration (at top of subroutine)

...

allocate(P(200), Q(100))                ! allocates P, Q (inside subroutine)

P = 0                                   ! fills P with all zeros

Q = 1                                   ! fills Q with all ones

P(101:200) = Q                          ! overwrites half of P with ones

Q = Q*P(1:100)                          ! whole array op (like Matlab’s .*)

deallocate(P, Q)                        ! frees the memory for P and Q

Allows code to look like normal math: the next-to-last line could instead be written
do i=1,100
   Q(i) = Q(i)*P(i)
end do

Very similar to Matlab, though multiplication and division are pointwise (not 
matrix-based).



To output values to the screen (or a file), you must use a print or write 
statement:

Fortran:                                           Matlab:  

write(*,*)                disp(‘  ‘)

print *, x                disp(x)

print *, ‘x = ‘,x         x  or  disp(sprintf(‘x = %g’,x))

Similar formatting arguments may be used as in Matlab’s sprintf command:

write(*,’(format string)’)   or   print ‘(format string)’

Here, format string  could be of the form  [Matlab equivalent]:  

Real numbers:  e15.3 [%15.3e], f15.7 [%15.7f], g15.7 [%15.7g]

Strings: a8 [%8s]

Integers:  i7 [%7i]

We can write to a file with ID 10 instead of standard output:
 double precision :: x                ! declare the variables
 ...                                  ! fill x with something interesting
 open(10, ‘outfile’)                  ! open file ‘outfile’ as stream ‘10’

 write(10,*) x                     ! write the value of x to stream ‘10’
 close(10)                            ! close the output file



We can easily read specified parameters from a file with a namelist:
 double precision :: a, b, c           ! declare the variables
 namelist / listname / a, b, c         ! declare the namelist
 ...
 open(100, ‘filename’, status=‘old’)   ! open the ‘filename’ as stream 100
 read(100, nml=listname)               ! read the namelist from stream 100
 close(100)                            ! close the file

Or we can read values from standard input:
integer :: n                            ! declare the variable

  ...
  print *, 'Enter n:’                     ! prompt the user for input
  read(*,*) n ! read in the variable

Or we can read values from a formatted input file:
 double precision :: a                 ! declare the variables
 ...
 open(20, ‘filename’, status=‘old’)   ! open the existing file ‘filename’
 read(20,*) a                         ! read the variable
 close(20)                            ! close the file



Matlab:

For loop (fwd/bwd):
for i=1:n
   ...
end
for i=n:-1:1
   ...
end

While loop:
while (condition)

...
end

Fortran:

 Do loop (fwd/bwd):
do i=1,n
   ...
end do
do i=n,1,-1
   ...
end do

 While loop:
do while (condition)
   ...
end do



Matlab:

If statements:
if (condition)
   ...
elseif (condition)
   ...
else
   ...
end

Fortran:

 If statements
if (condition) then
   ...
elseif (condition) then
   ...
else
   ...
endif



Fortran also has options that Matlab does not include, such as the ‘where’ 
statement that allows you to perform selective computation without 
requiring loops and ‘if’ statements:

real :: a(20,20)
...
where (a /= 0.0)  a = 1.0/a

or

real :: a(20,20)
where (a /= 0.0)

   a = 1.0/a

end where



Matlab:

Switch statements:

switch param

case 0
   ...

case 1

   ...

case 2
   ...

case {3, 5, 6, 7}

   ...
otherwise

   ...

end

Fortran:

Select statements:
select case (param)
case (0)
   ...
case (1)
   ...
case (2)
   ...
case (3, 5:7)
   ...
case default
   ...
end select



Other important control mechanisms  [Matlab equivalent]:

cycle:  skip the remainder of this loop iteration, and go to the next loop  
iteration  [continue]

exit:  exit the loop altogether  [break]

return:  exit the function or subroutine  [return]

stop:  terminates the program immediately  [quit]



subroutine foo(a)                ! name/args of subroutine

real*8 :: a(100)                 ! declare the arguments

integer :: i                     ! declare local variables

do i=1,100                       ! begin loop

   if (a(i) < 0) then            ! terminate program on negative value

      stop

   elseif (a(i) == 0.d0) then    ! skip zero-valued entries 

      cycle

   elseif (a(i) > 1.d20) then    ! stop loop if values too large

      exit

   else                          ! otherwise compute reciprocal

      a(i) = 1.d0/a(i)

   endif                         ! end of if statements

enddo                            ! end of do loop

return                           ! return to calling routine

end subroutine foo               ! end of subroutine



% Description: Computes the 

% dot-product of two vectors

% clear memory

clear

% declare variables to be used

n = 10000000;

a = zeros(n,1);

b = zeros(n,1);

i = 0;  

sum = 0;

stime = 0;

ftime = 0;

% initialize a and b

for i=1:n

  a(i) = 0.001*i/n;

  b(i) = 0.001*(n-i)/n;

end

% start timer
stime = cputime;

% compute dot-product

sum = 0.0;

for i=1:n

   sum = sum + a(i)*b(i);

end

% stop timer

ftime = cputime;

rtime = ftime-stime;

% output computed value and runtime

fprintf('dot-prod = %g',sum);

fprintf(' runtime = %g’, rtime);

% end program



program DotProd

  !-----------------------------------------

  ! Description: 

  ! Computes the dot-product of two vectors

  !-----------------------------------------

  !======= Declarations =========

  implicit none

  integer, parameter :: n=10000000

  integer :: i

  double precision, allocatable :: a(:), b(:)

  double precision :: sum, stime, ftime

  !======= Internals ============ 

  ! initialize the vectors

  allocate(a(n),b(n))

  do i=1,n

     a(i) = 1.d-3*i/n

     b(i) = 1.d-3*(n-i)/n

  enddo

  ! start timer
  call get_time(stime)

  ! compute dot-product
  sum = 0.d0
  do i=1,n,1
     sum = sum + a(i)*b(i)
  enddo

  ! stop timer
  call get_time(ftime)

  ! output computed value and runtime
  write(*,*) ' dot-product =',sum
  write(*,*) '     runtime =',ftime-stime

  ! free vectors
  deallocate(a,b)

end program DotProd
!========================================

Note: the ‘get_time’ function below is not built into Fortran (I’ll give it to you).



% Description: Computes pi via
% numerical integration
%   pi = 4*int_0^1 1/(1+x^2) dx
% using the midpoint rule over n 
% equal subintervals

% clear memory
clear

% declare variables to be used
n=0; i=0; h=0; x=0; f=0; a=0;
stime=0; ftime=0; rtime=0;
my_pi=0; err=0;

%  set integrand function handle
f = @(a) 4 / (1 + a*a);

% input number of intervals
n = input(...
  '# of intervals (0 quits):');
if (n < 1) 
   return
end

%  start timer
stime = cputime;

% set subinterval width
h = 1/n;

% perform integration
my_pi = 0;
for i=1:n
   x = h*(i - 0.5);
   my_pi = my_pi + h*f(x);
end
err = pi – my_pi;

% stop timer
ftime = cputime;
rtime = ftime-stime;

% output result and error
fprintf('  my pi = %.16e',my_pi);
fprintf('true pi = %.16e',pi);
fprintf('  error = %g',err);
fprintf('runtime = %g',rtime);
 



program ComputePi

  !--------------------------------------------------

  ! Description: 

  !    Computes pi through numerical integration via

  !        pi = 4*int_0^1 1/(1+x^2) dx

  !    We use a simple midpoint rule for integration, 

  !    over subintervals of fixed size 1/n, where n

  !    is a user-input parameter.

  !--------------------------------------------------

  !======= Declarations =========

  implicit none

  integer :: n, i

  double precision :: pi, h, x, f, a, stime, ftime

  double precision, parameter :: &

       pi_true = 3.14159265358979323846d0

  !======= Internals ============ 

  ! set the integrand function

  f(a) = 4.d0 / (1.d0 + a*a)

  ! input the number of intervals

  write(*,*) 'Enter number of intervals (0 quits):'

  read(*,*) n

  if (n < 1)  stop

  ! start timer

  call get_time(stime)

  ! set subinterval width

  h = 1.d0/n

  ! perform integration over n intervals

  pi = 0.d0

  do i=1,n,1

     x = h*(i - 0.5d0)

     pi = pi + h*f(x)

  enddo

  ! stop timer

  call get_time(ftime)

  ! output computed value and error

  write(*,*) ' computed pi =',pi

  write(*,*) '     true pi =',pi_true

  write(*,*) '       error =',pi_true - pi

  write(*,*) '     runtime =',ftime-stime

  

end program ComputePi

!===================================================



%====================================================

% Description: 

%    Computes equilibrium chemical densities at 

%    multiple spatial locations, given a (random) 

%    background temperature field.  The chemical rate 

%    equations and solution strategy are in the 

%    subroutine chem_solver, which is called at every

%    spatial location.

%====================================================

% clear all old variables from memory

clear

% declare variables to be used

n=0;  i=0;  its=0;  res=0;  maxit=0;  

lam=0;  err = 0;  stime=0;  ftime=0;

% set solver input parameters

maxit = 1000000;

lam = 1.d-2;

err = 1.d-10;

  

% input the number of intervals

n = input('Enter number of intervals (0 quits): ');

if (n < 1) 

   return

end

% allocate temperature and solution arrays

T = zeros(n,1);  u = zeros(n,1);

v = zeros(n,1);  w = zeros(n,1);

% set random temperature field, initial guesses at

% chemical densities

T = rand(size(T));      u = 0.35*ones(size(u));

v = 0.1*ones(size(v));  w = 0.5*ones(size(w));

% call solver over n intervals

stime = cputime;

for i=1:n

   [u(i),v(i),w(i),its,res] = ...     

      chem_solver(T(i),u(i),v(i),w(i),lam,err,maxit);

   if (res > err)

   disp(sprintf(‘error at step %i, res > err’,i))

      break

   end

end

ftime = cputime;

% output solution time

disp(sprintf('     runtime = %g',ftime-stime));

% end program 



program ChemicalEquilibrium

  !--------------------------------------------------

  ! Description: 

  !    Computes the equilibrium chemical densities at 

  !    multiple spatial locations, given a (random) 

  !    background temperature field.  The chemical 

  !    rate equations and solution strategy are in 

  !    the subroutine chem_solver, which is called at 

  !    every spatial location.

  !--------------------------------------------------

  !======= Declarations =========

  implicit none

  integer :: n, i, maxit, its

  real*8, allocatable :: T(:), u(:), v(:), w(:)

  real*8 :: lam, eps, res, stime, ftime

  !======= Internals ============

  ! set solver input parameters

  maxit = 1000000;  lam = 1.d-2;  eps = 1.d-10

  

  ! input the number of intervals

  write(*,*) 'Enter number of intervals (0 quits):'

  read(*,*) n

  if (n < 1)  stop

  ! allocate temperature and solution arrays
  allocate(T(n),u(n),v(n),w(n))

 

  ! set random temperature field, initial guesses 

  ! at chemical densities

  call random_number(T);  u = 0.35d0 

  v = 0.1d0;  w = 0.5d0

  ! call solver over n intervals

  call get_time(stime)

  do i=1,n,1

     call chem_solver(T(i),u(i),v(i),w(i),lam,eps, &

                      maxit,its,res)

     if (res > eps) then

        print *, ‘error at step’,i,’, res > eps’

  stop

     endif  

  enddo

  call get_time(ftime)

  ! output solution time

  write(*,*) '     runtime =',ftime-stime 

  ! free temperature and solution arrays

  deallocate(T,u,v,w)

end program ChemicalEquilibrium

!===================================================



function [u,v,w,its,res] = ... 
chem_solver(T,u,v,w,lam,err,maxit)

%====================================================

% Usage:

%   [u,v,w,its,res] = ...

           chem_solver(T,u,v,w,lam,err,maxit)

% Arguments:

%    T - (in), temperature

%    u - (i/o), concentration (in: guess, out: sol)

%    v - (i/o), concentration (in: guess, out: sol)

%    w - (i/o), concentration (in: guess, out: sol)

%  lam - (in), damping parameter (lambda)

%  err - (in), nonlinear solver tolerance

%  maxit - (in), maximum allowed iterations

%  its - (out), # required iterations for convergence

%  res - (out), final nonlinear residual (max norm)

%====================================================

% declare local variables for this function

k1 = 0;  k2 = 0;  k3 = 0;  k4 = 0;  f = zeros(3,1);

% compute chemical rate coefficients

k1 = exp(-5*T);

k2 = atan(5*(T-0.5))/3 + 0.5;

k3 = 1/cosh(5*(T-0.5));

k4 = tanh(5*(T-0.5)^2);

% compute initial residual function

f = chem_residual;  res = max(abs(f));

% loop

for its = 0:maxit

   % check for convergence

   if (res < err) 

      return

   end

   % fixed-point iteration solution update

   u = u + lam*f(1);

   v = v + lam*f(2);

   w = w + lam*f(3);

   % compute residuals

   f = chem_residual;  res = max(abs(f));

end

% declare the nested function chem_residual

   function resid = chem_residual

      resid(1) = k2*v + k1*u*(u+v+w-1);

      resid(2) = k1*u*(1-u-v-w) - (k2+k3)*v + k4*w;

      resid(3) = k3*v - k4*w;

   end

return

end

% end function



subroutine chem_solver(T,u,v,w,lam,eps,maxit,its,res)

  !--------------------------------------------------

  ! Arguments:

  !  T – r*8 (input), temperature

  !  u – r*8 (i/o), concentration (in:guess, out:sol)

  !  v – r*8 (i/o), concentration (in:guess, out:sol)

  !  w – r*8 (i/o), concentration (in:guess, out:sol)

  !  lam – r*8 (in), damping parameter (lambda)

  !  eps – r*8 (in), nonlinear solver tolerance

  !  maxit - int (in), maximum allowed iterations

  !  its - int (out), # iterations required

  !  res – r*8 (out), final residual (max norm)

  !--------------------------------------------------

  !======= Declarations =========

  implicit none

  integer, intent(in) :: maxit

  integer, intent(out) :: its

  double precision, intent(in) :: T, lam, eps

  double precision, intent(out) :: res

  double precision, intent(inout) :: u, v, w

  double precision :: k1, k2, k3, k4, f(3)

  !======= Internals ============ 

  ! compute chemical rate coefficients

  k1 = exp(-5.d0*T)

  k2 = atan(5.d0*(T-0.5d0))/3.d0 + 0.5d0

  k3 = 1.d0/cosh(5.d0*(T-0.5d0))

  k4 = tanh(5.d0*(T-0.5d0)**2)

  ! compute initial residual function

  f = chem_resid()

  res = maxval(abs(f))

  ! loop

  do its = 0,maxit

     ! check for convergence

     if (res < eps)  exit

     ! fixed-point iteration solution update

     u = u + lam*f(1)

     v = v + lam*f(2)

     w = w + lam*f(3)

     ! compute residuals

     f = chem_resid()

     res = maxval(abs(f))

  enddo

contains

  function chem_resid()

    double precision :: chem_resid(3)

    chem_resid(1) = k2*v + k1*u*(u+v+w-1.d0)

    chem_resid(2) = k1*u*(1.d0-u-v-w)-(k2+k3)*v+k4*w

    chem_resid(3) = k3*v - k4*w

  end function chem_resid

end subroutine chem_solver

!====================================================



Some typical errors in Fortran codes include:

Accidental declaration of constants with the wrong type:
   vol = 1/3*pi*r**3   ! some compilers will first compute 1/3=0, 
                       ! or will compute a single-precision
                       ! result instead of double-precision 

Since Fortran passes arguments by reference, subroutines will “trust” that your 
declarations are correct. Be certain to declare the types and sizes of subroutine 
arguments correctly.  A fix for this is an “interface block” (see book).

Be careful about punctuation, since all characters in a program are relevant.  A 
misplaced  ,  or  .  or  ;  or  :  could mean the difference between a functioning 
program, compiler errors, or more insidious run-time problems.

This is especially true for simple differences between Fortran and Matlab: 

for j=1:5

vs

do j=1,5

